
ПРИЛОЖЕНИЕ А 
Графический материал к аттестационной работе

Принципиальная схема изделия

DES алгоритм


Один раунд des алгоритма 

; roundr2 = 344 cycles
; roundr  = 324 cycles

roundr2:
	call	round
	call	keyrotr		; 20 cycles
	goto	keyrotr		; 20 cycles

roundr:
	call	round		; 302 cycles
	goto	keyrotr		;  20 cycles

; roundl2 = 338 cycles including call
; roundl  = 320 cycles including call

roundl2:
	call	keyrotl		; 18 cycles
roundl:
	call	keyrotl		; 18 cycles
; fall into round


; xxx cycles including call
;                 call =   2 cycles
;           clear newr =   4 cycles
; 2 sboxes @ 33 cycles =  66 cycles
; 6 sboxes @ 34 cycles = 204 cycles
;                  xor =   8 cycles
;                 swap =  16 cycles
;               return =   2 cycles
;                        ----------
;                total = 302 cycles

round:
; clear newr to make it easy to do the P-box permutation (by simply setting
; bits)
	clrf	newr+0
	clrf	newr+1
	clrf	newr+2
	clrf	newr+3	

; do s-box 1
	gethi	right+0,r32,r05			; 6 cycles
	xorkey	k14,k05,k17,k11,k24,k01		; 12 cycles
	corehi	sbox12,nr09,nr17,nr23,nr31	; 15 cycles

; do s-box 2
	getlo	right+0,r04,r09			; 6 cycles
	xorkey	k03,k10,k28,k15,k06,k21		; 12 cycles
	corelo	sbox12,nr13,nr28,nr02,nr18	; 15 cycles

; do s-box 3
	gethi	right+1,r08,r13			; 6 cycles
	xorkey	k23,k08,k19,k12,k04,k26		; 12 cycles
	corehi	sbox34,nr24,nr16,nr30,nr06	; 16 cycles

; do s-box 4
	getlo	right+1,r12,r17			; 6 cycles
	xorkey	k16,k02,k07,k27,k20,k13		; 12 cycles
	corelo	sbox34,nr26,nr20,nr10,nr01	; 16 cycles

; do s-box 5
	gethi	right+2,r16,r21			; 6 cycles
	xorkey	k41,k55,k52,k31,k37,k47		; 12 cycles
	corehi	sbox56,nr08,nr14,nr25,nr03	; 16 cycles

; do s-box 6
	getlo	right+2,r20,r25			; 6 cycles
	xorkey	k30,k48,k40,k51,k45,k33		; 12 cycles
	corelo	sbox56,nr04,nr29,nr11,nr19	; 16 cycles

; do s-box 7
	gethi	right+3,r24,r29			; 6 cycles
	xorkey	k44,k53,k49,k39,k56,k34		; 12 cycles
	corehi	sbox78,nr32,nr12,nr22,nr07	; 16 cycles

; do s-box 8
	getlo	right+3,r28,r01			; 6 cycles
	xorkey	k46,k32,k42,k50,k36,k29		; 12 cycles
	corelo	sbox78,nr05,nr27,nr15,nr21	; 16 cycles

; now exclusive-or the old left with the new right to complete the
; computation

        movf    left+0,W
	xorwf	newr+0
        movf    left+1,W
	xorwf	newr+1
        movf    left+2,W
	xorwf	newr+2
        movf    left+3,W
	xorwf	newr+3

; now swap the stuff around

        movf    right+0,W
	movwf	left+0
        movf    right+1,W
	movwf	left+1
        movf    right+2,W
	movwf	left+2
        movf    right+3,W
	movwf	left+3

        movf    newr+0,W
	movwf	right+0
        movf    newr+1,W
	movwf	right+1
        movf    newr+2,W
	movwf	right+2
        movf    newr+3,W
	movwf	right+3

	return

Для реализации раунда понадобились следующие макросы:

; The gethi and getlo macros implement six bit sections of the Expansion
; Permutation.

; 6 cycles
gethi   macro   rbyte,b20r,b20b,b10r,b10b
        swapf   rbyte,W
        andlw   0x0f
        btfsc   b20r,b20b
        iorlw   0x20
        btfsc   b10r,b10b
        iorlw   0x10
	endm

; 6 cycles
getlo   macro   rbyte,b20r,b20b,b10r,b10b
        movf    rbyte,W
        andlw   0x0f
        btfsc   b20r,b20b
        iorlw   0x20
        btfsc   b10r,b10b
        iorlw   0x10
	endm

; The xorkey macro is used to exclusive-or the 6-bit result of the Expansion
; Permuation in the W register (as generated by the gethi or getlo macros
; above) with the appropriate 6 bits of the current subkey (as defined by
; Permuted Choice 2).  The arguments are the six subkey bits to be used.

; 12 cycles
xorkey  macro   b20r,b20b,b10r,b10b,b08r,b08b,b04r,b04b,b02r,b02b,b01r,b01b
        btfsc   b20r,b20b
        xorlw   0x20
        btfsc   b10r,b10b
        xorlw   0x10
        btfsc   b08r,b08b
        xorlw   0x08
        btfsc   b04r,b04b
        xorlw   0x04
        btfsc   b02r,b02b
        xorlw   0x02
        btfsc   b01r,b01b
        xorlw   0x01
	endm

; The corehi and corelo macros are used to perform the S-box lookup and
; the P-box permutation.    The only difference between the two macros is
; which half of the sbox table value is used.  The corehi macro must be
; used for S-boxes 1, 3, 5, and 7, while the corelo macro must be used for
; S-boxes 2, 4, 6, and 8.

; The first argument to the macro is the entry point of the S-box table to
; be used.

; The second through fifth arguments to the macro are the bits that the
; P-box permutes the S-box outputs into, in order from the most to the least
; significant.

; 15 cycles for sbox12
; 16 cycles for others
corehi  macro   sbox,b7r,b7b,b6r,b6b,b5r,b5b,b4r,b4b
	call	sbox			; 6 cycles for sbox12,
					; 7 cycles for others
	movwf	temp
	btfsc	temp,7
        bsf     b7r,b7b
	btfsc	temp,6
        bsf     b6r,b6b
	btfsc	temp,5
        bsf     b5r,b5b
	btfsc	temp,4
        bsf     b4r,b4b
	endm

; 15 cycles for sbox12
; 16 cycles for others
corelo  macro   sbox,b3r,b3b,b2r,b2b,b1r,b1b,b0r,b0b
	call	sbox			; 6 cycles for sbox12,
					; 7 cycles for others
	movwf	temp
	btfsc	temp,3
        bsf     b3r,b3b
	btfsc	temp,2
        bsf     b2r,b2b
	btfsc	temp,1
        bsf     b1r,b1b
	btfsc	temp,0
        bsf     b0r,b0b
	endm
Генерация ключа

setkey:	movlw	keylen
	movwf	temp

setkey0:
        rlf     INDF
	rrf	keyl+0
        rlf     INDF
	rrf	keyl+1
        rlf     INDF
	rrf	keyl+2
        rlf     INDF
	rrf	keyl+3
        rlf     INDF
	rrf	keyr+2
        rlf     INDF
	rrf	keyr+1
        rlf     INDF
	rrf	keyr+0

        incf    FSR
	decfsz	temp
	goto	setkey0

        swapf   keyl+3,W        ; move low 4 bits of keyl+3
	movwf	keyr+3		;   into high 4 bits of keyr+3

	movlw	0f0h		; mask off extra bits
	andwf	keyl+3
	andwf	keyr+3

        movlw   - keylen             ; restore FSR to original value
        addlw   FSR

	return

